

# EVALUATING UNTYPED $\lambda$ -EXPRESSIONS IN HASKELL

October 4, 2023

SUPERVISOR: Prof. Alessi Fabio

AUTHOR: D'Ambrosi Denis

CO-SUPERVISOR: Prof. Di Gianantonio Pietro

Student ID: 147681

## ABSTRACT

The recent widespread of Generative AI has caused a frantic turmoil of both excitement and apprehension all around the world. This technology has been adopted by many influential multinational companies, possibly directly or indirectly impacting billions of users globally, but there is still no shared consensus about the best guidelines to follow when developing or utilizing this form of innovation. In this paper, we will provide an examination of the most impelling risk scenarios that need to be addressed and will present an overview about the most relevant research efforts on possible countermeasures.

## CONTENTS

<b>1</b>	<b>Pure <math>\lambda</math>-Calculus</b>	<b>2</b>
<b>2</b>	<b>Evaluating <math>\lambda</math>-expressions</b>	<b>3</b>
2.1	Lexing $\lambda$ -expressions . . . . .	3
2.2	Parsing $\lambda$ -expressions . . . . .	4
2.3	Evaluating $\lambda$ -terms: the algebraic approach . . . . .	6
2.3.1	Alpha conversion method . . . . .	6
2.3.2	De Bruijn Indexes method . . . . .	8
2.4	Evaluating $\lambda$ -terms: the computational approach . . . . .	10
<b>3</b>	<b>Conclusions</b>	<b>12</b>

## INTRODUCTION

In this report, we will explore various approaches to normalize expressions belonging to Church's  $\lambda$ -Calculus. In Section 1, we briefly introduce the syntax and semantics of  $\lambda$ -Calculus, along with its reduction techniques. In Section 2 we will present the implementations of three different strategies to solve this problem. Finally, in Section 3 we will analyze the pros and cons of each approach and present some potential future work.

### 1. PURE $\lambda$ -CALCULUS

Pure  $\lambda$ -Calculus, introduced by Alonzo Church in the 1930s [Chu36], is a formal system that plays a pivotal role in the foundation of theoretical computer science. It is a minimalist, yet immensely powerful, mathematical framework for expressing computation using only anonymous functions.

The set of  $\lambda$ -terms  $\Lambda$  is the minimal set that can be constructed inductively according to the following rules (that effectively define the syntax of this formalism):

1. If  $x$  is a variable, then  $x \in \Lambda$
2. If  $x$  is a variable and  $M \in \Lambda$ , then  $(\lambda x.M)$
3. If  $M, N \in \Lambda$ , then  $(M N) \in \Lambda$

Rule number 2 expresses the possibility of creating *abstractions*. An abstraction is an anonymous function that takes as input a  $\lambda$ -term (in this case  $x$ ) and returns a  $\lambda$ -term (not necessarily a variable, in this case  $M$ ) as output. Multiple input functions can be constructed via a process called *currying*: since higher-order-functions are allowed within  $\lambda$ -Calculus, we can transform any  $n$ -ary function into an abstraction that takes only one input and returns another abstraction as output. The latter will, in turn, consume only one input and produce yet another abstraction, and so on,  $n$  times. Evaluating the last function will, eventually, produce the desired output.

Rule number 3, on the other hand, allows for the application of abstractions to  $\lambda$ -terms: performing such an operation will yield a new  $\lambda$ -term according to two main reduction operations:

1.  $\alpha$ -conversion aims at renaming bound variables for avoiding name collisions:  $(\lambda x.M) \mapsto (\lambda y.M[x := y])$
2.  $\beta$ -reduction normalizes  $\lambda$ -terms by replacing bound variables in the body of the abstraction with the actual parameter:  $((\lambda x.M)N) \mapsto M[x := N]$

Before executing a  $\beta$ -reduction step, we want to ensure that the actual parameter substituted within an abstraction's body does not appear *free* in it. Formally, provided a  $\lambda$ -term  $t$ , the set of its *Free variables*  $FV(t)$  can be computed recursively as follows:

- $FV(x) = x$  if  $x$  is a variable
- $FV((\lambda x.M)) = FV(M) \setminus x$
- $FV((M N)) = FV(M) \cup FV(N)$

If the input of the application appears free within the abstraction, then we must execute a process of  $\alpha$ -conversion on the latter to avoid any undesired change of semantics of the function caused by name clashes. If we did not take this additional care while substituting, we may inadvertently compute incorrect terms while  $\beta$ -reducing.

This quick overview of the syntax and semantics of the  $\lambda$ -Calculus has laid the ground-work for introducing the actual implementations of  $\lambda$ -expressions evaluators of the upcoming sections. Before continuing, it is worth discussing briefly the reason why this formalism is still relevant, even after all these years. In 1937, Turing proved that  $\lambda$ -Calculus is Turing complete [Tur37], meaning that the formalism has the computational power equivalent to a universal Turing machine. This critical property establishes  $\lambda$ -Calculus as a foundational formal system in the study of computation: such result implies that any computable function can be expressed and evaluated within the confines of  $\lambda$ -Calculus. The consequences of this fact underpin its applications not only in theoretical computer science but also in practical domains, where it serves as the basis for functional programming languages and aids in the development of algorithms for various computational tasks.

## 2. EVALUATING $\lambda$ -EXPRESSIONS

*A monad is just a monoid in the category of endofunctors*

---

SAUNDERS MAC LANE[ML71]

Our objective consists in the implementation of an evaluator for  $\lambda$ -expressions that computes the normal form of each given  $\lambda$ -term. To provide a user-friendly interface to our software, we allow users to write the input  $\lambda$ -expressions into a text file, that gets lexed and parsed before applying the reduction strategies. Our technology stack is completely Haskell-based and is structured as follows:

1. Alex [DJM] as lexer generator
2. Happy [GM] as parser generator
3. The evaluation strategy is implemented in pure Haskell with some support from the Prelude, List and Control modules

In the following sections, we will give a brief overview of each phase by commenting the key pieces of code needed to perform the relative computation. Please note that the actual evaluation strategies are explained in detail in sections 2.3 and 2.4

### 2.1. LEXING $\lambda$ -EXPRESSIONS

As previously introduced, our lexer generator of choice is Alex. Alex allows developers to define a set of regular expression-based rules to recognize lexemes from a text stream and produce the relative tokens according to predetermined actions. In practice, we first need to declare the set of our syntactic tokens by defining an abstract data type with a constructor for each single token. In our case, we need to recognize variables (that can be modelled as alphabetical strings), the  $\lambda$  symbol, dot, round parentheses, whitespaces and newlines characters (to allow the user to specify multiple expressions). In practice, this straightforwardly translates in the following Haskell datatype:

```

1 data Token = NewLine
2   | Term String
3   | Dot
4   | Lambda
5   | OpenPar
6   | ClosedPar
7 deriving (Eq, Show)
```

Now we can actually recognize lexemes through regular expressions. Alex’s syntax allows users to associate names to regexes before use to improve readability: in our case, it will be helpful to define the language of single letters (both lower and uppercase):

```
1 $letter = [a-zA-Z]
```

We can then use the above expression in the lexemes-recognition rules:

```
1 tokens :-
2     \n          { \s -> NewLine }
3     $letter+    { \s -> Term s }
4     \.          { \s -> Dot }
5     \λ          { \s -> Lambda }
6     \(         { \s -> OpenPar }
7     \)         { \s -> ClosedPar }
8     $white      ;
```

Note that:

1. We have associated an action to each lexeme in the form of a (ironically) Haskell  $\lambda$ -expression that maps the read string into the relative token. In our case, the actions were trivially defined since we are dealing with a very simple language, but there is no limit to the complexity of these functions. Supporting functions can be also defined along with the custom datatype.
2. The lexemes are ordered according to descending priority: even though the newline character will be recognized by both the first and the last expression, only the first one will be applied to produce a `NewLine` token.

By compiling the file through Alex, we obtain a custom Haskell module that can be called from other modules to read any text stream and recognize its lexemes. The following parser module will exploit such functionality to convert the input file into a stream of tokens that can be then converted into an abstract syntax tree.

## 2.2. PARSING $\lambda$ -EXPRESSIONS

Following a very similar computational pipeline to compilers’ frontends, the lexed tokens must be analyzed by a parser. In our case, we adopted the Happy parser generator, which, given a grammar specified in Backus-Naur Form, produces the Haskell code to parse it. The modules produced through Happy adopt a *LARL(1) bottom-up strategy* which is less powerful than the more general *LR(1)* approach, yet can be implemented more efficiently. Since diving more deeply into the expressiveness and computational complexity of parsing techniques is out of the scope of this report, the reader can be satisfied by knowing that:

1. Happy-generated parsers are expressive enough to recognize  $\lambda$ -Calculus terms (as a matter of fact, they are enough expressive to work in the GHC frontend!)
2. In the only case of this project where we would have benefited from a top-down parsing approach see the closure issue in Section 2.4, we easily overcame the problem by firstly building a syntax tree in a bottom-up fashion and then traversing the structure with a top-down algorithm.

The objective of our parser will be to recognize the syntactic structure of each  $\lambda$ -expression and to feed it as a reasonable data structure to the evaluator. A trivial data type suitable for the task can be defined as follows:

```

1 data Term =
2   Var String
3   | Abs String Term
4   | App Term Term
5   | Empty
6   deriving (Eq)

```

This abstract data type straightforwardly mirrors the construction of the set of  $\lambda$ -terms  $\Lambda$  introduced in Section 1. The `Empty` constructor is declared only to streamline the handling of empty lines in the input file and will not be considered in the following evaluating strategies. As we can see, `Term` derives the standard `Eq` type class, but not `Show`'s one. Instead, we decided to implement a custom function to pretty-print functions:

```

1 instance Show Term where
2   show :: Term -> String
3   show (Var x) = x
4   show (Abs x t) = join ["(λ", x, ".", show t, ")"]
5   show (App x y) = join ["(", show x, " ", show y, ")"]
6   show _ = error "Empty detected"

```

We can perform another step to improve the readability of the parser's specification: to each token (and token's data field) defined in Section 2.1, we can associate a short name to avoid unnecessary verbosity later on.

```

1 %token
2   NL      { NewLine }
3   VAR     { Term $$ }
4   '.'     { Dot }
5   'λ'     { Lambda }
6   '('     { OpenPar }
7   ')'     { ClosedPar }

```

We are finally ready to formally describe the language's grammar. Yet again, following Section's 1 set definition it is almost effortless to write a comprehensive description of  $\lambda$ -Calculus expressions: the only additional care we must take is to handle correctly multiple expressions in the same input file.

```

1 start  : line                { [$1] }
2        | start line         { $2 : $1 }
3
4 line   : NL                  { Empty }
5        | exp NL              { $1 }
6
7 exp    : VAR                  { Var $1 }
8        | '(' exp ')'         { $2 }
9        | 'λ' VAR '.' exp     { Abs $2 $4 }
10       | exp exp              { App $1 $2 }

```

Before proceeding with the description of the evaluation strategies, it is worth to analyze two details about this definition:

1. Our specification requires the input stream to end with a `NewLine` token: we could solve this issue by introducing additional symbols and/or rewriting rules, but disambiguating the obtained grammar would require a lot of effort. Instead, we opted to "artificially" append a `NewLine` token at the end of the lexed stream before passing it to the parsing procedure, circumventing the issue efficiently and painlessly.

2. Contrary to the construction defined in Section 1, our specification allows the user to omit some brackets. This language expansion is a source of ambiguity, but we can tackle it by defining the abstraction and application as left-associative operations through the `%left 'λ' '.' VAR` directive. In the end, we are still left with a few shift/reduce conflicts, but thankfully the defaulting rules of Happy produce a parser with the desired behaviour without additional work.

The generated parser will be the actual program compiled/interpreted, thus we need to define the main function of the software:

```

1  main :: IO ()
2  main = do
3      s <- readFile "../test.txt"
4      let tokens = alexScanTokens s ++ [ NewLine ]
5      let parsedTerms = filter (/= Empty) (reverse (parse tokens))
6      let evaluatedTerms = map eval parsedTerms
7      mapM_ (print) evaluatedTerms

```

Let us continue by diving deeper into multiple possible implementations of the `eval` procedure.

### 2.3. EVALUATING $\lambda$ -TERMS: THE ALGEBRAIC APPROACH

In this section we will introduce an algebraic approach to the problem of normalizing  $\lambda$ -terms: at its core, this strategy implements the  $\beta$ -reduction operation described in Section 1 to substitute terms within a suitable data structure. As previously explained, with this approach problems may arise due to involuntary name capture of free variables while normalizing, so we will elaborate on two different methods to overcome this issue.

#### 2.3.1. ALPHA CONVERSION METHOD

The first approach implements the  $\alpha$ -conversion operation introduced in Section 1. For starters, let us illustrate the code of the `eval` function:

```

1  eval :: Term -> Term
2  eval (App (Abs x t) t1) = let t1' = eval t1 in eval (sub x t t1')
3  eval (App t1 t2) = case eval t1 of
4      a@(Abs x t1) -> eval (App a t2)
5      t -> App t (eval t2)
6  eval (Abs x t) = Abs x (eval t)
7  eval Empty = error "Empty detected"
8  eval t = t

```

The process of  $\beta$ -reduction is applied in the first pattern of the definition: if a term consists in an application of an abstraction of variable  $x$  and body  $t$  to a term  $t_1$ , then substitute each (non-shadowed) occurrence of  $x$  in  $t$  with the normal form of  $t_1$ . The substitution routine `sub` will be introduced shortly. On the other hand, if the root of the input data structure is an application of a non-abstraction term to another term, normalize the first argument: if it reduces to an abstraction, apply recursively the first pattern, otherwise return an application of the two normalized arguments. Finally, the remaining patterns are required to evaluate recursively abstractions and to provide a base-case for the recursion.

Actually substituting a term within a term requires a slightly more complex algorithm:

```

1 sub :: String -> Term -> Term -> Term
2 sub x v@(Var y) newVal
3   | x == y = newVal
4   | otherwise = v
5 sub x a@(Abs y t) newVal
6   | x == y = a
7   | notElem y (freeVars newVal) = Abs y (sub x t newVal)
8   | otherwise = let y' = firstNameAvailable (freeVars newVal) in Abs y' (sub x (
9     ↪ alphaConv y t y') newVal)
9 sub x (App t1 t2) t = App (sub x t1 t) (sub x t2 t)
10 sub _ Empty _ = error "Empty detected"

```

The first pattern is the one that actually performs the substitution: if we have a variable denominated with the same symbol that must be substituted, then the variable is swapped-out in favor of the input term `newVal`, otherwise the expression is unchanged. This ensures that, for example,  $x[x := y] \mapsto y$  and  $x[z := y] \mapsto x$ .

The second pattern actually introduces the  $\alpha$ -reduction, as it deals with the substitution of a term within an abstraction. Let us explain the three cases separately:

1. If  $x = y$ , then it means that the symbol to substitute has been shadowed by a new declaration, thus the substitution must not be spread further. By doing that, we can guarantee that

$$\begin{aligned}
&(x(\lambda x.zx))[x := (\lambda y.y)] \mapsto ((\lambda y.y)(\lambda x.zx)) \\
&(x(\lambda x.zx))[x := (\lambda y.y)] \not\mapsto ((\lambda y.y)(\lambda x.z(\lambda y.y)))
\end{aligned}$$

2. The variable bound by the abstraction does not appear free within the input term, so we can apply the substitution without worrying about name clashes. Note that the function `freeVars` will be soon discussed.
3. In this case,  $y \in FV(a)$ , so we must apply the  $\alpha$ -conversion procedure to obtain an abstraction that bounds another symbol (that does not belong in  $FV(a)$ ), but with the same semantics. After renaming the variable, we can substitute recursively the applied parameter to the body of the updated function.

Finally, the third case only propagates the substitution through both children of an application. The last pattern only deals with potential misuse of the `eval` function (remember that we filtered out all the `Empty` objects right after parsing).

The specification of `freeVars` follows diligently the definition of  $FV$  presented in Section 1:

```

1 freeVars :: Term -> [String]
2 freeVars (Var x) = [x]
3 freeVars (Abs x t1) = removeDuplicates (freeVars t1) \\ [x] where
4   removeDuplicates :: [String] -> [String]
5   removeDuplicates [] = []
6   removeDuplicates (x:xs)
7     | elem x xs = removeDuplicates xs
8     | otherwise = x : removeDuplicates xs
9 freeVars (App t1 t2) = freeVars t1 ++ freeVars t2
10 freeVars _ = error "Empty detected"

```

Where `(\\)` is the list removal operator imported from `Data.List`.

Before concluding, we still need to explain how we engineered the  $\alpha$ -conversion process. The renaming function (`alphaConv`) traverses the structure in a top-down fashion while substituting each occurrence of the problematic variable (clearly also the  $\lambda$ -bindings) with the new symbol.

```

1  alphaConv :: String -> Term -> String -> Term
2  alphaConv x v@(Var y) z
3      | x == y = Var z
4      | otherwise = v
5  alphaConv x (Abs y t) z
6      | x == y = Abs z (alphaConv x t z)
7      | otherwise = Abs y (alphaConv x t z)
8  alphaConv x (App t1 t2) z = App (alphaConv x t1 z) (alphaConv x t2 z)
9  alphaConv _ Empty _ = error "Empty detected"

```

We still need to find a new, suitable, name. This task is executed by the function `firstNameAvailable`, called in a previous code snippet.

```

1  firstNameAvailable :: [String] -> String
2  firstNameAvailable = firstNameAvailableRec 0 where
3      firstNameAvailableRec :: Int -> [String] -> String
4      firstNameAvailableRec n usedVars
5          | elem (variableSet !! n) usedVars = firstNameAvailableRec (n+1) usedVars
6          | otherwise = variableSet !! n

```

This procedure returns the first occurrence of a variable belonging to `variableSet`, but not to `usedVars`. From a Software Engineering perspective, one may be concerned by the fact that there is no error handling code: if all the variables within `variableSet` appeared in `usedVars`, we would end up trying to access an unexistent location of the former. Thankfully, Haskell features a lazy-evaluation mechanism that allows for the definition of infinite structures. As a matter of fact, `variableSet` has been defined as a list, ordered by length (and, subsequently, alphabetically), of all the possible strings (thus we could visualize it as `["a", "b", "c", ..., ↦ "z", "aa", "ab" ..., "zz", "aaa", ...]`): in practice, this guarantees that a new symbol will always be found.

```

1  variableSet :: [String]
2  variableSet = concatMap (\k -> replicateM k ['a'..'z']) [1..]

```

Note that `replicateM` is the function imported from `Data.List` and `[1..]` is the infinite set of positive integers.

### 2.3.2. DE BRUIJN INDEXES METHOD

An alternative to  $\alpha$ -conversion is implementing the De Bruijn Indexes. This consists of translating the  $\lambda$ -expressions into an alternative syntax where the variables bound by abstractions are not identified by their symbol, but instead by the level of nesting of their binding. There are multiple ways of representing free variables, but we chose to use directly the symbol with which they appeared in the original expression. Some examples of this conversion are as follows:

$$\begin{aligned}
 (\lambda x.x) &\mapsto (\lambda 1) \\
 (\lambda x.c) &\mapsto (\lambda c) \\
 (\lambda x.(\lambda y.(\lambda z.(x\ z)c))) &\mapsto (\lambda(\lambda(\lambda((3\ 1)c))))
 \end{aligned}$$



By applying this conversion, we eliminate the requirement to bind names, so we can apply term substitutions within abstractions based solely on the order of application. To implement this strategy, we clearly need a new data type that re-defines variables and abstractions:

```

1  data DBTerm = DBVar Int
2    | DBConst String
3    | DBAbs DBTerm
4    | DBApp DBTerm DBTerm
5    deriving (Eq)
6
7  instance Show DBTerm where
8    show :: DBTerm -> String
9    show (DBVar n) = Prelude.show n
10   show (DBConst x) = x
11   show (DBAbs t) = join ["(λ", show t, ")"]
12   show (DBApp t1 t2) = join ["(", show t1, " ", show t2, ")"]

```

And we need a procedure capable of handling the translation:

```

1  toDeBruijn :: Term -> DBTerm
2  toDeBruijn = toDeBruijnRec [] where
3    toDeBruijnRec :: [String] -> Term -> DBTerm
4    toDeBruijnRec vars (Var x) = case elemIndex x vars of
5      Just n -> DBVar (n+1)
6      Nothing -> DBConst x
7    toDeBruijnRec vars (Abs x t) = DBAbs (toDeBruijnRec (x:vars) t)
8    toDeBruijnRec vars (App t1 t2) = DBApp (toDeBruijnRec vars t1) (toDeBruijnRec vars
9      ↪ t2)
10   toDeBruijnRec _ Empty = error "Empty detected"

```

This function traverses recursively the tree that defines the expression, gradually keeping track of the bound variables in the parameter `vars`. This ensures that, when we actually encounter a term in the form `(Var x)`, we can establish whether `x` was a bound name (if there exists at least an index in correspondence of which the symbol appears in `vars`), or if it is a constant, thanks to the monadic characterization of `elemIndex`. Note that we managed `vars` as a stack instead of a list (we append items to the head, using the `(:)` operator) to allow name shadowing: by doing this, if `elemIndex` finds an integer, we are sure that it is the innermost binding of that variable.

After the translation, normalizing  $\lambda$ -terms in De Bruijn's notation becomes quite easier compared to the algebraic method with  $\alpha$ -conversion. In fact, while the `eval` procedure remains practically the same, the `sub` routine is far less intricate, without the need of supporting functions:

```

1  eval :: Term -> DBTerm
2  eval t = dbEval (toDeBruijn t) where
3    dbEval :: DBTerm -> DBTerm
4    dbEval (DBApp (DBAbs t) t1) = let t1' = dbEval t1 in dbEval (sub t t1' 1)
5    dbEval (DBApp t1 t2) = case dbEval t1 of
6      a@(DBAbs t) -> dbEval (DBApp a t2)
7      t -> DBApp t (dbEval t2)
8    dbEval (DBAbs t) = DBAbs (dbEval t)
9    dbEval t = t
10
11  sub :: DBTerm -> DBTerm -> Int -> DBTerm
12  sub (DBVar n) t n'
13    | n == n' = t

```

```

14     | otherwise = DBVar n
15   sub c@(DBConst _) _ _ = c
16   sub (DBAbs t1) t2 n = DBAbs (sub t1 t2 (n+1))
17   sub (DBApp t1 t2) t n = DBApp (sub t1 t n) (sub t2 t n)

```

The `sub` function substitutes a variable with a given term only if the level of nesting of the variable is equal to the one of the searched binding (note that when we recursively iterate on an abstraction, we increase such number by 1, as we are "moving away" from the relevant binding). Clearly, constants never get substituted (notice how here we do not have the issue of name clashes with the free variables) and applications only need to spread the substitution through the tree's structure.

#### 2.4. EVALUATING $\lambda$ -TERMS: THE COMPUTATIONAL APPROACH

While the  $\alpha$ -conversion and the De Bruijn Indexes method are very similar at their core (firstly we guarantee to avoid name clashes through some form of transformation of the expression and then we apply a substitution), the computational method takes a completely different approach. It aims at exploiting Haskell's anonymous functions to implement actual  $\lambda$ -abstractions capable of evaluating terms.

First of all, we need to define a new abstract data type capable of storing variables, applications and anonymous functions:

```

1  data CompTerm = CompVar String
2    | CompAbs (CompTerm -> CompTerm)
3    | CompApp CompTerm CompTerm

```

Since anonymous functions do not belong to the type classes **Eq** and **Show**, the whole data type will not be able to derive them either automatically. While being able to check terms for equality is not something we will need, printability is clearly crucial to ensure that we computed the right normal forms. We will explain our solution to this problem at the end of the section.

At first glance, this approach may seem very straightforward, as Haskell supports higher-order anonymous functions by default. The hidden issue in this technique lies in the closures: while we could manually write a function like  $(\lambda x.(\lambda y.x))$  and exploit the language's  $\lambda$ -expression functionality to evaluate it, there is no straightforward way to nest recursively an unbounded number of expressions: in the previous example, we would not be able to build the innermost term as the constant  $x$ , as we would not know where said symbol was bound. Instead, we can implement our own elementary environments as lists of associations between names and parameters and use them to define recursively the  $\lambda$ -expressions:

```

1  eval :: Term -> CompTerm
2  eval = evalRec [] [] where
3    evalRec :: [CompTerm] -> [String] -> Term -> CompTerm
4    evalRec env vars (Abs v t) = CompAbs (\ x -> evalRec (x:env) (v:vars) t)
5    evalRec env vars (Var x) = case elemIndex x vars of
6      Just n -> env !! n
7      Nothing -> CompVar x
8    evalRec env vars (App t1 t2) = let t2' = evalRec env vars t2 in case evalRec env
9      ↪ vars t1 of
10        CompAbs f -> f t2'
11        v@(CompVar _) -> CompApp v t2'
12        a@(CompApp _ _) -> CompApp a t2'
13    evalRec _ _ Empty = error "Empty detected"

```

Let us dissect the above code snippet to better understand how it works. The recursive procedure takes three parameters as input:

1. A list of actual parameters (initially empty)
2. A list of names for said parameters (initially empty)
3. An algebraic term to traverse

If the term is an abstraction, we create a new `CompAbs` instance that holds an actual anonymous function. Said function will be created recursively thanks to the environment augmented with the actual parameter (in this case, `x`) and the formal parameter (in this case, `v`). Note that, again, we build the closure as a stack to allow name shadowing later on.

If the term is a variable, then it must mean that we must distinguish whether it is bound or free. To do so, we must check if it appears within `vars`: if it does, then we can return the corresponding term saved in the closure `env`; if not, that means that it is a constant and can be returned as is.

If the term is an application, then we have to check the type of its first argument: if it is an abstraction, then we can straightforwardly apply it to the normalized parameter  $t'_2$ . On the other hand, if it is an application or a variable, we can always reduce the second argument and then save the normalized application.

By substituting the algebraic terms with actual functions we avoid the issue of name clashes, since when an abstraction is ready for application, it will evaluate its argument according to the environment indexing, in a similar fashion to De Bruijn's Indexes.

As introduced before, printing an expression will not be as simple as in the previous approaches: it is not possible to inspect an anonymous function's body after creation, which essentially becomes a non-examinable *black-box*. Our best bet is to deduce the structure of a  $\lambda$ -expression by passing selected parameters to it and looking at how it computes the result: we can not scrutinize the computation steps by themselves, but the instances of `CompVar` can. For example, if we had a function  $f$  and we knew that  $fa = a$ , then we could infer that either  $f = \lambda x.x$  or  $f = \lambda x.a$ . We must find a way to discriminate between such cases: by coming back to the definition of our lexer in Section 2.1, we can see that we did not allow numbers as identifiers, only alphabetic characters. If we fed numerical arguments (as strings, of course), to the functions, we could easily distinguish between actual bound variables (that would end up being numerical) and constants (which will belong to the set of recognized tokens, i.e. letters). This procedure can be applied recursively to deduce the structure of the abstractions at each nesting level.

In practice, we coded a function that transforms a `CompTerm` into a `Term` by feeding strings encoding numbers incrementally larger into the abstractions and observing the results:

```

1  toSyntax :: CompTerm -> (Term, [String])
2  toSyntax = toSyntaxRec 0 where
3      toSyntaxRec :: Int -> CompTerm -> (Term, [String])
4      toSyntaxRec _ (CompVar x) = case readMaybeInt x of
5          Just n -> (Var (show n), [])
6          Nothing -> (Var x, [x])
7      toSyntaxRec n (CompAbs f) = let (term, freeVars) = toSyntaxRec (n+1) (f (CompVar (
8          ↪ show n))) in (Abs (show n) term, freeVars)
9      toSyntaxRec n (CompApp t1 t2) = let
10          (t1', freeVars1) = toSyntaxRec n t1
11          (t2', freeVars2) = toSyntaxRec n t2
12          in (App t1' t2', removeDuplicates (freeVars1 ++ freeVars2)) where

```

```

12         removeDuplicates :: [String] -> [String]
13         removeDuplicates [] = []
14         removeDuplicates (x:xs)
15             | elem x xs = removeDuplicates xs
16             | otherwise = x : removeDuplicates xs
17
18 readMaybeInt :: String -> Maybe Int
19 readMaybeInt = readMaybe

```

In the base case, `toSyntaxRec` discriminates between bound (values that can be read as an integer) and free (values that encode strings) variables. Starting from this, it constructs the set of free variables with a top-down approach: applications only need to spread the conversion further and functions must feed a number and increment it gradually as explained before. In the end, we compute a pair that contains the algebraic expression as the first coordinate and the set of free variables as the second.

Now we only need to substitute the numeric symbols with alphabetical names while being careful of not to cause accidental name captures. The following function does exactly that:

```

1  substInts :: (Term, [String]) -> Term
2  substInts (term, freeVars) = substIntsRec [] term freeVars where
3      substIntsRec :: [String] -> Term -> [String] -> Term
4      substIntsRec varNames (Var x) _ = case readMaybeInt x of
5          Just n -> Var (varNames !! n)
6          Nothing -> Var x
7      substIntsRec varNames (App t1 t2) usedVars = App (substIntsRec varNames t1
8          ↪ usedVars) (substIntsRec varNames t2 usedVars)
9      substIntsRec varNames (Abs _ t) usedVars = let x = firstNameAvailable usedVars in
10         ↪ Abs x (substIntsRec (varNames ++ [x]) t (usedVars ++ [x]))
11      substIntsRec _ Empty _ = error "Empty detected"

```

We substitute the bound variables and the bindings while being careful to avoid choosing the free variables (note the use of the previously defined function `firstNameAvailable`).

### 3. CONCLUSIONS

In this report, we implemented different techniques to evaluate  $\lambda$ -expressions. The algebraic approach is generally simpler to realize, especially in Haskell where efficient recursion and pattern matching make the implementation of the substitution procedure almost trivial. This approach may suffer from the problem of name clashes: to avoid it, we analyzed both the  $\alpha$ -conversion and the De Bruijn Indexes techniques. The former requires some care to handle the names, but yet again Haskell's capability of handling infinite structures through lazy evaluation proves to be useful as we have access to an unlimited set of symbols. The latter, on the other hand, requires a conversion at first, but the substitution becomes effortless after. We decided to keep the expressions in De Bruijn's notation to provide some examples of this alternative, but it would be easy to write a function that translates the terms back into the original notation if we wanted to. Finally, we showed how it is possible to normalize expressions exploiting Haskell's own anonymous functions: this approach, although seems easy at first, quickly becomes slightly cumbersome as it requires creating environments and developing a quite complicated print mechanism with multiple conversions.

As future work, it would be interesting to see if there are simpler, more elegant alternatives to the double-translation printing procedure for the computational approach and, more

generally, if it is possible to tackle the problem of normalizing  $\lambda$  expressions with alternative methods.

#### REFERENCES

- [Chu36] Alonzo Church. An unsolvable problem of elementary number theory. *American Journal of Mathematics*, 58:345, 1936.
- [DJM] Chris Dornan, Isaac Jones, and Simon Marlow. Alex user guide. <https://haskell-alex.readthedocs.io/en/latest/about.html>.
- [GM] Andy Gill and Simon Marlow. Happy documentation. <https://haskell-happy.readthedocs.io/en/latest/index.html>.
- [ML71] Saunders Mac Lane. *Category Theory for the Working Mathematician*. Springer, 1971.
- [Tur37] Alan M. Turing. Computability and  $\lambda$ -definability. *The Journal of Symbolic Logic*, pages 153–163, 1937.